

# Numerical Contour Integration

## Applied to One- and Two-Dimensional Integrals over Distributions

**Erickson Tjoa**

We present a straightforward implementation of contour integration by setting options for `Integrate` and `NIntegrate`, taking advantage of powerful results in complex analysis. As such, this article can be viewed as documentation to perform numerical contour integration with the existing built-in tools. We provide examples of how this method can be used when integrating analytically and numerically some commonly used distributions, such as Wightman functions in quantum field theory. We also provide an approximating technique when time-ordering is involved, a commonly encountered scenario in quantum field theory for computing second-order terms in Dyson series expansion and Feynman propagators. We believe our implementation will be useful for more general calculations involving advanced or retarded Green's functions, propagators, kernels and so on.

### ■ Introduction

It is well known that we can integrate analytically a large class of functions with known anti-derivatives via `Integrate`; otherwise, we can use `NIntegrate` for numerical results. There are various settings that one can use to evaluate integrals, depending on the task at hand. Crucially, this can be performed even if the integrand is complex valued, such as the function  $e^{ix}$ . In some cases, when the integrand is a distribution, this integration can also be done analytically by setting the option `PrincipalValue`, for example, or by imposing a regulator, which in physics is often called a UV cutoff.

In this article, we are interested in a contour integral of the form

$$\int_{\gamma} F(z) dz, \tag{1}$$

where  $F(z)$  is a complex-valued function or distribution and  $\gamma$  is an integration contour, with possibly higher-dimensional generalizations. For a large class of functions  $F(z)$  and a contour  $\gamma$ , this can be done in various ways, such as using powerful techniques in complex analysis like Cauchy's integral formula or the residue theorem. For many of these functions, `Integrate` can provide answers immediately, sometimes involving special functions.

The motivation for this article is based on the observation that there is no good documentation on how to perform explicit contour integration in Mathematica. It turns out that the ingredients necessary to perform this task are already available within the software. We believe that some of these ingredients are already used elsewhere for different purposes. Our task is to synthesize these components in a coherent way to document how to perform numerical contour integration, using only built-in commands and their options. We will show by way of examples that in many cases this method proves superior to direct integration with  $i\epsilon$  prescription, especially when the integrand is complicated or when the integral is multidimensional.


## ■ Example 1: Simple Textbook Example—One-Dimensional Contour Integral with Closed-Form Solutions

Consider the integral

$$I_1 := \lim_{\epsilon \rightarrow 0^+} \int_{\mathbb{R}} \frac{e^{ikx}}{x - i\epsilon} dx = \pi i + \text{PV} \int_{\mathbb{R}} \frac{e^{ikx}}{x} dx = 2\pi i, \quad (2)$$

where  $\epsilon > 0$ ,  $k > 0$ , and  $\gamma$  is the entire real line. The first equality comes from Sokhotsky's formula [1] and the second equality uses principal value integration. This is a commonly used "epsilon-regularization" in physics when evaluating certain apparently divergent integrals, which has to do with the distributional nature of the integrand in question (see Appendix A in [1] for examples).

This integral can be solved using complex analysis instead of Sokhotsky's formula, based on the observation that  $I_1$  can be computed using the following contour.



The contour is the horizontal blue segment and the semi-circular arc  $\gamma$ . The integral over the semi-circular arc vanishes as  $R \rightarrow \infty$  by Jordan's lemma [2, 3]. The contribution from the closed loop can be calculated in various ways, such as the residue theorem. The residue theorem says that since the closed loop encloses the pole at  $x = i \epsilon$  (using the standard counterclockwise convention [2, 3]),  $I_1 = 2 \pi i \text{Res}(e^{ikx} / (x - i \epsilon)) = e^{-\epsilon k}$ . Taking the limit as  $\epsilon \rightarrow 0$ , we get the same result  $2 \pi i$  as before.

There is indeed a built-in function for calculating residues.

$$\text{Residue} \left[ \frac{\text{Exp}[\mathbf{I} \mathbf{k} \mathbf{x}]}{\mathbf{x} - \mathbf{I} \epsilon}, \{\mathbf{x}, \mathbf{I} \epsilon\} \right]$$

$$e^{-k \epsilon}$$

Another way to evaluate  $I_1$  is based on the fact that the “ $i \epsilon$  prescription” is an instruction to perform a specific contour integration (without an  $\epsilon$ -regulator) in the sense that

$$I_1 = \int_{\gamma} \frac{e^{ikx}}{x} dx = 2 \pi i \text{Res} \left( \frac{e^{ikx}}{x}, 0 \right) = 2 \pi i, \tag{3}$$

where  $\gamma$  is a contour chosen to go from  $-\infty$  to  $\infty$  but deformed near the pole  $x = 0$  to the lower complex plane.

The  $i \epsilon$  prescription in (2) shifts the pole at  $x = 0$  to  $x = i \epsilon$  in the upper half-plane, so that we can integrate over the real line and then take the limit as  $\epsilon \rightarrow \infty$ . Consequently, if we were to remove the  $\epsilon$ -regulator, the deformation theorem [2] would require that we deform the contour to the lower complex plane near  $x = 0$  in order to obtain the same value of the integral. (The deformation theorem [2] in complex analysis states that a contour integral remains constant under deformation of the contour that does not cross any poles or branch cuts.) For this contour, the residue is instead given by the following.

$$\text{Residue} \left[ \frac{\text{Exp}[\mathbf{I} \mathbf{k} \mathbf{x}]}{\mathbf{x}}, \{\mathbf{x}, \mathbf{0}\} \right]$$

1

If we choose  $\epsilon < 0$  instead, the integral is zero. Here is an explicit computation for fixed  $k = 2$ .

### □ Case 1

$$\mathbf{I1@a\_} := \text{Integrate}\left[\frac{\mathbf{Exp}[2 \mathbf{I} \mathbf{x}]}{\mathbf{x} - \mathbf{I} \mathbf{a}}, \{\mathbf{x}, -\infty, \infty\}\right]$$

$$\mathbf{I1/@}\{1/10, 1/100, -1/10, -1/100\}$$

$$\left\{\frac{2 \mathbf{i} \pi}{e^{1/5}}, \frac{2 \mathbf{i} \pi}{e^{1/50}}, 0, 0\right\}$$

The zero values have a simple explanation in terms of contour integration: since the pole is now shifted to the lower half-plane, the contour does not enclose any poles, which by Cauchy's integral theorem guarantees that the integral is zero.

### □ Case 2

This is a good place to introduce the numerical contour integration scheme. As a simple example, we can use a rectangular contour.

$$\mathbf{I2[k\_ , a\_]} := \text{Integrate}\left[\frac{\mathbf{Exp}[\mathbf{I} \mathbf{k} \mathbf{x}]}{\mathbf{x}}, \{\mathbf{x}, -\infty, -\mathbf{a}, -\mathbf{a} - \mathbf{I}, \mathbf{a} - \mathbf{I}, \mathbf{a}, \infty\}, \mathbf{Assumptions} \rightarrow \mathbf{k} > \mathbf{0}\right]$$

$$\mathbf{I2@@@}\{\{1, 1\}, \{\pi, 1\}, \{\pi, 1/3\}\}$$

$$\{2 \mathbf{i} \pi, 2 \mathbf{i} \pi, 2 \mathbf{i} \pi\}$$

### □ Case 3

For any  $k > 0$  and  $a > 0$ , this calculation gives the correct answer, as expected from direct calculation using residues or textbook pen-and-paper calculations (see e.g. [3]). Varying  $a$  does not change the integral, as a consequence of the deformation theorem. In particular, other contours also work, such as, for example, a triangular contour.

$$\mathbf{I3[k\_ , a\_]} := \text{Integrate}\left[\frac{\mathbf{Exp}[\mathbf{I} \mathbf{k} \mathbf{x}]}{\mathbf{x}}, \{\mathbf{x}, -\infty, -\mathbf{a} \mathbf{I}, \infty\}, \mathbf{Assumptions} \rightarrow \mathbf{k} > \mathbf{0}\right]$$

```
I3 @@@ {{1, 1}, {1, 1/3}}
{2 i π, 2 i π}
```

#### □ Case 4

For completeness, let us show that in this simple case the  $i\epsilon$  prescription (which is commonly used in physics) can also be done. We add the condition  $a > 0$ .

```
I4[k_, a_] := Integrate[Exp[I k x]
  / (x - I a), {x, -∞, ∞},
  Assumptions -> k > 0 && a > 0]

I4[1, a]

2 i e-a π
```

The final result involves  $e^{-a}$ , which tends to 1 as  $a \rightarrow 0^+$ . Therefore, in this simple example, all the methods yield good results. It is straightforward to show that many common textbook examples (see e.g. [3]) can be performed in this way.

#### □ Case 5

Finally, our main goal is to show that `NIntegrate` can also perform the task reliably. First do the  $i\epsilon$  prescription.

```
I5[k_, a_] :=
  Quiet@NIntegrate[Exp[I k x]
    / (x - I a), {x, -∞, ∞},
    WorkingPrecision -> 30]

I5[1, 1/100]

0. × 10-29 + 8.64997542233668860083260374951 i
```

Here we see the first instance where the  $i\epsilon$  prescription starts to fail: when performed numerically, the most basic setting does not work, and increasing the working precision does not help much.

Remark: The `Quiet` command silences all error messages and is used for readability. In practice, in order to discover the issues and possibly infer the sources of problems, it is useful to not use `Quiet` on a first try.

## □ Case 6: Numerical Contour Integration

Now do the contour integral using a triangular contour.

```

I6[k_, a_] :=
  Quiet@NIntegrate[ $\frac{\mathbf{Exp}[\mathbf{I k x}]}{\mathbf{x}}$ , {x, -Infinity, -a I, Infinity}]

I6@@@ {{1, 1}, {1, 1 / 10}}

{0. + 6.28319 i, 4.44089 × 10-16 + 6.28319 i}

```

The contour integration works as is without further modification. The size of the contour does not matter, as long as it is not so small that it leads to numerical instabilities associated with small numbers.

We refer to the numerical computation using `NIntegrate` that involves contour deformation in the complex plane as numerical contour integration. We will see recurring examples where  $i\epsilon$  prescriptions, albeit simplest to implement, almost never work when needed in spite of optimizing various settings. Thus numerical contour integration is a highly competitive technique to use.

## ■ Example 2: Two-Dimensional Integral on $\mathbb{R}^2$ over a Distribution

(Remark: Some of the discussion here was briefly outlined in [4, 5], which is based on what this article contains.)

The main example we consider in this article involves the integral over a distribution or bi-distribution of the form

$$J_1(\Omega, \sigma) := \lim_{\epsilon \rightarrow 0^+} \int_{\mathbb{R}} \int_{\mathbb{R}} e^{\frac{-t^2}{2\sigma^2}} e^{\frac{-T^2}{2\sigma^2}} e^{-i\Omega(t-T)} W(t, \mathbf{x}; T, \mathbf{x}; \epsilon) dt dT, \quad (4)$$

where  $\Omega \in \mathbb{R}$ ,  $\sigma > 0$ ,  $\epsilon > 0$  and  $W(t, T, \epsilon)$  is a distribution or bi-distribution given by

$$W(t, \mathbf{x}; T, \mathbf{X}; \epsilon) := -\frac{1}{4\pi^2} \frac{1}{(t-T-i\epsilon)^2 - (\mathbf{x}-\mathbf{X})^2}. \quad (5)$$

In physics, more specifically quantum field theory, this distribution is the two-point vacuum correlation function (also called the two-point Wightman function) of a massless scalar field in  $(3+1)$ -dimensional Minkowski space [6].

$$W(t, \mathbf{x}; T, \mathbf{X}; \epsilon) = \langle 0 | \phi(t, \mathbf{x}) \phi(T, \mathbf{X}) | 0 \rangle \quad (6)$$

The arguments of the two-point function are the coordinates of two spacetime events  $(t, \mathbf{x})$  and  $(T, \mathbf{X})$ , where  $\mathbf{x} = (x, y, z)$  and  $\mathbf{X} = (X, Y, Z)$  in Cartesian coordinates and  $\epsilon$ . The integral  $J_1(\Omega, \sigma)$  is evaluated at  $\mathbf{x} = \mathbf{X}$  and is thus independent of the spatial coordinates. The  $i\epsilon$  prescription tells us that the Wightman function is a distribution or bi-distribution [6].

Our task is to evaluate the integral  $J_1(\Omega, \sigma)$  in (4). This is a two-dimensional complex integral over  $t$  and  $T$  with a continuum of poles at  $t = T + i\epsilon$  for any fixed  $\epsilon > 0$ . In the literature,  $J_1(\Omega, \sigma)$  gives the transition probability (divided by a small coupling constant  $\lambda^2$ ) of an Unruh–DeWitt (UDW) detector consisting of a two-level quantum system (qubit) with energy gap  $\Omega$  interacting with the massless scalar field for a finite duration prescribed by the Gaussian function with timescale set by  $\sigma$  (see e.g. [7] and references therein). The Gaussian functions in the integrand ensure that the interaction between the detector and the quantum field  $\phi$  is adiabatically switched on and off smoothly; that is, the Fourier transform of the Wightman function has polynomial tails and is strongly suppressed by the exponential tails of the Fourier transform of the Gaussians. This guarantees that there is no spurious divergence in the transition probability due to switching on the detector very sharply or suddenly, such as with a switching function with discontinuity at an endpoint, like a rectangular function, which is unphysical. An analogy is switching on a lightbulb: in realistic scenarios the current in the circuit does not simply increase from zero to a constant value instantaneously, but rather smoothly over some short time interval.

The example in equation (4) is important for three reasons.

1. It is one of the simplest examples in quantum field theory; many other examples in physics involve distributions that take similar and sometimes more complicated forms (e.g. different power laws  $(t - T - i\epsilon)^n$  where  $n \neq 2$  or  $(t - f(T) - i\epsilon)^2$  for some function  $f$ ).
2. It is one of the simplest examples of a two-dimensional integral with a continuum of poles, thus going beyond standard textbook examples (e.g. in [3] all the examples are one dimensional).
3. Despite its somewhat complicated appearance, it admits a closed-form expression [7].

#### □ Benchmark Value for $J_1$

In [7] it is shown that the exact closed-form expression is given by

$$J_1(\Omega, \sigma) = \frac{1}{4\pi} \left( e^{-\sigma^2 \Omega^2} - \Omega \sigma \pi^{1/2} \operatorname{erfc}(\Omega \sigma) \right), \quad (7)$$

where  $\operatorname{erfc}(x)$  is the complementary error function. We want to show that numerically this result can be obtained in a satisfactory way. For this purpose, let us focus on a particular numerical value by setting  $\Omega \sigma = 1$ , so that

$$J_1(\Omega \sigma = 1) \simeq 0.0078827. \quad (8)$$

We can then use this value as a benchmark for our methods.

```
J1[Ω_, σ_] :=  $\frac{1}{4 \pi}$  (Exp[-Ω2 σ2] - Sqrt[π] Ω σ Erfc[Ω σ])

J1[Ω, 1/Ω] // N

0.00708827
```

## □ Direct $i \epsilon$ Integration

Direct  $i \epsilon$  integration for  $J_1$  is given by the following command.

```
J2[Ω_, σ_, ε_, tmax_, minrecursion_] :=
 $-\frac{1}{4 \pi^2}$  NIntegrate[Exp[- $\frac{t^2}{2 \sigma^2}$ ] × Exp[- $\frac{T^2}{2 \sigma^2}$ ] × Exp[-I Ω (t - T)]
 $\frac{1}{(t - T - I \epsilon)^2}$ , {t, -tmax, tmax}, {T, -tmax, tmax},
MinRecursion → minrecursion] // Quiet
```

The most modest settings yield poor results.

```
Module[
{Ω = 1, σ = 1, tmax = ∞, minrecursion = 3},
TableForm[
ParallelTable[{10.0-ε, J2[Ω, σ, 10-ε, tmax, minrecursion]}],
{ε, 0, 3}]]
]

1.      0.00424507 - 2.3529 × 10-14 i
0.1     0.16067 + 2.67381 × 10-17 i
0.01    0.703251 + 0.190412 i
0.001   1.91608 - 2.1148 × 10-15 i
```

Clearly, although the real parts are all positive (hence physically at least sensible for a transition probability), the results are very different from the benchmark value. The third value is wrong because the probabilities should be real (hence numerically the imaginary part should be much closer to zero).



Increasing `MinRecursion` as a first remedy does not help much, if at all.

```
Module[
  { $\Omega = 1$ ,  $\sigma = 1$ ,  $tmax = \infty$ ,  $minrecursion = 6$ },
  TableForm[
    ParallelTable[{ $10.0^{-\epsilon}$ , J2[ $\Omega$ ,  $\sigma$ ,  $10^{-\epsilon}$ ,  $tmax$ ,  $minrecursion$ ]},
      { $\epsilon$ , 0, 3}]]
  ]

1.      0.00424507 + 0. i
0.1     0.00244826 - 0.0000929703 i
0.01    0.149596 + 0.00266373 i
0.001   2.63054 + 0.0931637 i
```

## □ Restricting the Domain

Restricting the integration domain to  $(-5\sigma, 5\sigma)$ , where the Gaussian is strongly supported, improves the result, but not enough. (The negligible part of the integral associated to Gaussian tails was ignored.) To see that the behavior scales badly with  $\epsilon$ , let us choose a particular setting.

```
Module[
  { $\Omega = 1$ ,  $\sigma = 1$ ,  $tmax = 5$ ,  $minrecursion = 3$ },
  TableForm[
    ParallelTable[{ $10.0^{-\epsilon}$ , J2[ $\Omega$ ,  $\sigma$ ,  $10^{-\epsilon}$ ,  $tmax$ ,  $minrecursion$ ]},
      { $\epsilon$ , 0, 5}]]
  ]

1.      0.00424507 +  $9.28979 \times 10^{-15}$  i
0.1     0.00670271 -  $2.23593 \times 10^{-16}$  i
0.01    0.00704838 -  $8.47189 \times 10^{-16}$  i
0.001   -0.732931 -  $9.03524 \times 10^{-7}$  i
0.0001  -6.84952 - 1.52192 i
0.00001 -27.3218 + 0.246437 i
```

The first two entries, while not accurate enough, are again at least physically valid results because the imaginary part is negligible and hence can represent transition probabilities. In contrast, the last three values have negative real parts and are thus invalid by default.

We leave it to the reader to verify that other possible settings, such as increasing `WorkingPrecision` or `AccuracyGoal`, do not help, nor does changing the `Method` setting from `GlobalAdaptive` to another. There may be some combination of settings that makes this integration work, but if they exist then they are for our purposes not worth

the time. We will in fact show that the exact same setting that yields the bad results above does work with numerical contour integration.

A possible reason for these issues is that  $i\epsilon$  prescription changes the asymptotic behavior at  $t, T = \pm\infty$ . Numerically, we are evaluating the integral from  $t = -\infty - i\epsilon$  to  $t = \infty + i\epsilon$ , so `NIntegrate` has to deal with points at infinity that are shifted with a very small imaginary component.


## □ Numerical Contour Integration

Here  $J_1$  is transformed into a numerical contour integral,  $J_3$ . In contrast to the previous subsection, the  $\epsilon$ -regulator now serves as a contour deformation near a reasonably shaped contour. Hence the asymptotic behavior of the integral is not altered by varying  $\epsilon$ .

$$J_3(\Omega, \sigma) := \int_{\mathbb{R}} \int_{\gamma(\epsilon)} e^{\frac{-t^2}{2\sigma^2}} e^{\frac{-T^2}{2\sigma^2}} e^{-i\Omega(t-T)} W(t, \mathbf{x}; T, \mathbf{x}; 0) dt dT, \quad (9)$$

where  $\gamma(\epsilon)$  is a contour along the real line but deformed around the poles of the Wightman function. Here we have  $\epsilon = 0$  in the Wightman function because we have converted the  $i\epsilon$  prescription into an instruction to perform contour integration, so the  $\epsilon$  is now part of the definition of the contour  $\gamma$ .

Here is the contour  $\gamma(\epsilon)$ , where  $T = t$  is the location of the continuum of poles for every fixed  $t$ .



Since many contours work, let us first choose a particularly simple one: a rectangular deformation near the poles. If we integrate over  $T$  first, then the continuum of poles is located in the complex plane at  $t = T$ . Therefore, the contour  $\gamma$  should be deformed to the upper complex plane.

```
J3[Ω_, σ_, ε_, tmax_, minrecursion_] :=
- 1 / (4 π^2) NIntegrate[Exp[-t^2 / (2 σ^2)] × Exp[-T^2 / (2 σ^2)] × Exp[-I Ω (t - T)]
1 / (t - T)^2, {t, -tmax, tmax},
{T, -tmax, -tmax + t - ε, -tmax + t - ε + I, -tmax + t + ε + I,
-tmax + t + ε, tmax}, MinRecursion → minrecursion] // Quiet
```

Based on the previous subsection, we see that using  $t_{\max} = 5\sigma$  yields a better answer, since more than 99.99% of the area of the Gaussian in the integrand is contained in the interval  $(-5\sigma, 5\sigma)$ . We will not set  $t_{\max} = \infty$  in what follows. Consequently, the parts of the contour from  $-\infty$  to  $-5\sigma$  and  $5\sigma$  to  $\infty$  have been truncated and do not yield significant error. This gives the following results.

```
Module[{Ω = 1, σ = 1, tmax = 5, minrecursion = 3},
  TableForm[
    ParallelTable[{10.0-ε, J3[Ω, σ, 10-ε, tmax, minrecursion]}],
    {ε, -1, 3}]]
]
```

```
10.      0.00708827 + 2.7874 × 10-12 i
1.       -118.349 + 0.0394913 i
0.1     -39.6743 - 0.00597572 i
0.01    -63.5535 + 0.0846851 i
0.001   -92.5487 + 0.113553 i
```

This result may look undesirable, but the first value is numerically the same as the benchmark value up to very small imaginary part. This suggests that in the neighborhood of  $\epsilon \approx 10$  there is a good range that we can use to get a numerically accurate result; we will show that the result is invariant under smooth deformation of the contour. For example, if we zoom in between  $\epsilon \approx 10^1$  and  $\epsilon \approx 10^2$ , we get the following results.


```
Module[{Ω = 1, σ = 1, tmax = 5, minrecursion = 3},
  TableForm[
    ParallelTable[{ε, J3[Ω, σ, ε, tmax, minrecursion]}],
    {ε, 10, 50, 5}]]
]
```

```
10      0.00708827 + 2.7874 × 10-12 i
15      0.00708827 + 2.30492 × 10-12 i
20      0.00708827 + 2.01536 × 10-13 i
25      0.00708827 + 4.3746 × 10-12 i
30      0.00708827 + 1.21304 × 10-12 i
35      0.00708827 - 2.3751 × 10-12 i
40      0.00708827 + 1.23849 × 10-13 i
45      0.00708827 - 1.92872 × 10-12 i
50      0.00708827 + 2.79498 × 10-12 i
```

That the results are unchanged as we vary  $\epsilon$  (only the negligible error changes a little) is what we expect from the deformation theorem in complex analysis, since the deformation does not cross any singularities. This invariance under contour deformation is very useful because it is one of the consistency checks we can do (or perhaps need to do) in the absence of exact closed-form expressions and other cross-validation. Direct  $i\epsilon$  integration does not allow this, because every  $\epsilon$  yields different values (because the asymptotics are also shifted by  $i\epsilon$ ). Furthermore, numerical contour integration is also remarkably fast.

## □ A Different Contour

Here is a different contour. The width is fixed but the height toward the imaginary axis varies.



```
J4[Ω_, σ_, ε_, tmax_, minrecursion_] :=
- 1/(4 π^2) NIntegrate[Exp[-t^2/(2 σ^2)] × Exp[-T^2/(2 σ^2)] × Exp[-I Ω (t - T)]
  1/(t - T)^2, {t, -tmax, tmax},
  {T, -tmax, -tmax + ε I, tmax + ε I, tmax},
  MinRecursion → minrecursion] // Quiet

Module[{Ω = 1, σ = 1, tmax = 5, minrecursion = 3},
  TableForm[
    ParallelTable[{10.0^-ε, J4[Ω, σ, 10^-ε, tmax, minrecursion]},
      {ε, -1, 5}]]
]
```

10.	$1.78612 \times 10^7 + 3.83932 \times 10^6 i$
1.	$0.00708827 - 2.63646 \times 10^{-19} i$
0.1	$0.00708827 - 3.77893 \times 10^{-18} i$
0.01	$0.00708827 - 5.03938 \times 10^{-16} i$
0.001	$-0.73201 + 0.0000108048 i$
0.0001	$-6.84891 + 1.52209 i$
0.00001	$-27.3215 + 0.246435 i$

This choice of contour highlights that while contour integration is much more powerful than the direct  $i\epsilon$  integration method, one should avoid numerical issues associated with values of  $\epsilon$  that are either too large or too small; sometimes different contours yield different stability. Nonetheless, the main takeaway is that we have found that there is a large enough range of  $\epsilon$  for which the contour can be varied but still yields the correct value (in this case,  $\epsilon \approx 10^{-2} - 1$ ). In this range, we can vary  $\epsilon$  and the answer is invariant. The correct range of  $\epsilon$  that provides reliable results must however be found empirically and depends strongly on the integrand. For completeness, we show the results for zooming into the range of  $\epsilon$  (except that this time we need this to be  $\epsilon \approx 0.01 - 1$ ).

```
Module[{Ω = 1, σ = 1, tmax = 5, minrecursion = 3},
  TableForm[
    ParallelTable[{N[ε], J4[Ω, σ, ε, tmax, minrecursion]},
      {ε, 10-1, 1, 10-1}]
  ]
```

```
0.1    0.00708827 - 3.77893 × 10-18 i
0.2    0.00708827 + 8.7223 × 10-18 i
0.3    0.00708827 - 3.60317 × 10-18 i
0.4    0.00708827 - 1.06076 × 10-16 i
0.5    0.00708827 - 8.36374 × 10-15 i
0.6    0.00708827 + 2.94409 × 10-15 i
0.7    0.00708827 + 2.23166 × 10-15 i
0.8    0.00708827 - 3.47624 × 10-15 i
0.9    0.00708827 + 2.98312 × 10-15 i
1.     0.00708827 - 2.63646 × 10-19 i
```

A significant advantage of numerical contour integration over direct  $i\epsilon$  integration is that it allows for robust consistency check: if we can find a range for  $\epsilon$  in which the integral is constant (up to small numerical errors like a small imaginary part), then the integral is likely to be correct. This is an extremely important necessary condition, especially when a closed-form expression is not available. Of course, if there are other ways to check the correctness of the value (as in this example where we know the exact closed-form expression or if there are physical arguments to back it up), we should always also perform these other checks.

### ■ Example 3: Two-Dimensional Integral over a Distribution Containing the Heaviside Step Function

Next, we consider a more complicated integral,

$$K_1(\Omega, \sigma) := -\lim_{\epsilon \rightarrow 0^+} \int_{\mathbb{R}} \int_{\mathbb{R}} e^{\frac{-t^2}{2\sigma^2}} e^{\frac{-T^2}{2\sigma^2}} e^{i\Omega(t+T)} (\theta(t-T) W(t, \mathbf{x}; T, \mathbf{X}; \epsilon) + \theta(T-t) W(T, \mathbf{X}; t, \mathbf{x}; \epsilon)) dt dT, \quad (10)$$

where  $\Omega \in \mathbb{R}$ ,  $\sigma > 0$ ,  $\epsilon > 0$ ,  $\theta(z)$  is the Heaviside step function and the two-point Wightman function  $W(t, \mathbf{x}; T, \mathbf{X}; \epsilon)$  is given in equation (5). This integral differs conceptually from Example 2 in two respects:

- (1) the additional Heaviside step function in the integrand
- (2) we also allow for  $\mathbf{x} \neq \mathbf{X}$  and changed the sign on the phase

The Heaviside step function appearing in the integrand naturally arises in physics calculations where a Dyson series expansion is involved (i.e. where the notion of time ordering is necessary). In particular, this occurs in various time-dependent perturbation theory calculations within quantum mechanics and quantum field theory. The specific integral in (10) appears as part of the calculation of the nonlocal part of the joint two-detector density matrix of two qubits interacting with a massless scalar field in the so-called entanglement harvesting protocol [7, 8].

Perhaps somewhat surprisingly, this integral admits a closed-form solution [7], given by

$$K_1(\Omega, \sigma) = \frac{i\sigma}{4\pi^{1/2}L} e^{-\Omega^2\sigma^2 - \frac{L^2}{\sigma^2}} \left( 1 + \operatorname{erf}\left(i\frac{L}{\sigma}\right) \right), \quad (11)$$

where  $L = |\mathbf{X} - \mathbf{x}|$ . As before, we can assume that the coordinate system is aligned so that  $\mathbf{X} - \mathbf{x} = (X - x, 0, 0) = (L, 0, 0)$ .

#### □ Benchmark Value for $K_1$

For benchmarking, assume  $\Omega\sigma = 1$  and set  $L = \sigma$ . This gives

$$K_1(\Omega\sigma = 1, L = \sigma) \approx -0.0248507 + 0.0404108 i. \quad (12)$$

Here is the calculation.

$$\mathbf{K1}[\Omega\_ , \sigma\_ , L\_ ] := \frac{\mathbf{I} \sigma}{4 \mathbf{Sqrt}[\pi] L} \mathbf{Exp}\left[-\Omega^2 \sigma^2 - \frac{L^2}{4 \sigma^2}\right] \left(1 + \mathbf{Erf}\left[\frac{\mathbf{I} L}{2 \sigma}\right]\right)$$

$$\mathbf{K1}[1, 1, 1.0]$$

$$-0.0248507 + 0.0404108 i$$

This is an alternative if one prefers arbitrary  $\sigma$  in the expression.

$$\mathbf{K1}[\sigma^{-1}, \sigma, \sigma] // \mathbf{N}$$

$$-0.0248507 + 0.0404108 \, i$$

This number is complex, so we no longer have the benefit of verifying our numerical calculation by requiring that the imaginary part be small (which we used in Example 2).

Let us now evaluate  $K_1(\Omega, \sigma)$  numerically. We will not bother with the  $i \epsilon$  prescription anymore because it does not work; this is not surprising since the Heaviside step function only complicates the integral compared to Example 2. However, we will evaluate this integral in two ways, both using numerical contour integration, with the two methods differing in how they handle the Heaviside step function.

□ **Absorbing the Heaviside Function into the Upper Limit**


We rewrite the integral  $K_1$  as:

$$K_2(\Omega, \sigma) := - \int_{-\infty}^{\infty} \int_{-\infty}^t e^{\frac{-t^2}{2\sigma^2}} e^{\frac{-T^2}{2\sigma^2}} e^{i\Omega(t+T)} W(t, \mathbf{x}; T, \mathbf{X}; \epsilon) dT dt -$$

$$\int_{-\infty}^{\infty} \int_{-\infty}^T e^{\frac{-t^2}{2\sigma^2}} e^{\frac{-T^2}{2\sigma^2}} e^{i\Omega(t+T)} W(T, \mathbf{X}; t, \mathbf{x}; \epsilon) dt dT$$
(13)

By suitable relabeling, the two integrals are equal (this property is used in [7, 8]), but we will not need this fact, since in more general situations the two integrals are different. (See e.g. [5, 9], where the upper limits differ slightly by a “relative redshift” factor.) Although equation (13) is written in the notation of an  $i \epsilon$  prescription, this is merely a shorthand for performing the numerical contour integration, since we already know that direct  $i \epsilon$  integration does not work.

Pick a rectangular contour and set it so that  $\mathbf{X} - \mathbf{x} = (X - x, 0, 0) = (L, 0, 0)$ . For this choice, the two integrals have a continuum of poles at  $T = t - L$  and  $t = T - L$ , respectively. The contour will be chosen so that it does not cross any poles as  $\epsilon$  varies.



We can then write the following integral command.

```

K2[ $\Omega$ _,  $\sigma$ _,  $\epsilon$ _,  $x$ _,  $X$ _,  $tmax$ _,  $minrecursion$ _] :=
-Quiet[NIntegrate[Exp[- $\frac{t^2}{2\sigma^2}$ ]  $\times$  Exp[- $\frac{T^2}{2\sigma^2}$ ]  $\times$  Exp[ $I\Omega(t+T)$ ]
  ( $-\frac{1}{4\pi^2} \frac{1}{(t-T)^2 - (x-X)^2}$ ), { $t$ , - $tmax$ ,  $tmax$ },
  { $T$ , - $tmax$ , - $tmax + \epsilon I$ ,  $t + \epsilon I$ ,  $t$ },
  MinRecursion  $\rightarrow$   $minrecursion$ ]] -
Quiet[NIntegrate[Exp[- $\frac{t^2}{2\sigma^2}$ ]  $\times$  Exp[- $\frac{T^2}{2\sigma^2}$ ]  $\times$  Exp[ $I\Omega(t+T)$ ]
  ( $-\frac{1}{4\pi^2} \frac{1}{(T-t)^2 - (X-x)^2}$ ), { $T$ , - $tmax$ ,  $tmax$ },
  { $t$ , - $tmax$ , - $tmax + \epsilon I$ ,  $T + \epsilon I$ ,  $T$ },
  MinRecursion  $\rightarrow$   $minrecursion$ ]]]

```

The most modest settings yield good results.

```

Module[{ $\Omega = 1$ ,  $\sigma = 1$ ,  $L = 1$ ,  $tmax = 5$ ,  $minrecursion = 3$ },
TableForm[
  ParallelTable[
    { $10.0^{-\epsilon}$ , K2[ $\Omega$ ,  $\sigma$ ,  $10^{-\epsilon}$ , 0,  $L$ ,  $tmax$ ,  $minrecursion$ ]},
    { $\epsilon$ , -1, 5}]]]

```

10.	-55 229.3 - 1.26222 $\times 10^7$ <i>i</i>
1.	-0.0248507 + 0.0404108 <i>i</i>
0.1	-0.0248507 + 0.0404108 <i>i</i>
0.01	-0.0248507 + 0.0404108 <i>i</i>
0.001	-0.0248507 + 0.0404108 <i>i</i>
0.0001	-0.0246624 + 0.0414845 <i>i</i>
0.00001	-0.0287788 + 0.0299646 <i>i</i>

The contour parameter  $\epsilon$  is good in the range  $(10^{-3}, 1)$  and starts to deviate at around  $10^{-4}$ .

Remarkably, we can compute this result very well despite the extra complications, with nearly no additional settings beyond **MinRecursion**  $\rightarrow$  3.



□ **Approximate Heaviside Step Function**

The integral in equation (9) is actually written in a more useful form than the upper limit form in equation (12), in the sense that it accommodates a more general scenario where the two-point functions may not be simple functions of  $t$  and  $T$ . In general, the poles of the Wightman functions for the problem at hand can be located at  $t = f(T)$ , where  $f$  is some function that does not have a simple inverse. In such a situation, the upper limit of the integrals is not simply  $t$  or  $T$  and may have to be worked out numerically.

More specifically, in the context of quantum field theory in curved spacetimes, the Wightman function in equation (5) is given in flat spacetime in terms of the Minkowski coordinates, and the integrals we computed thus far correspond to two Unruh–DeWitt (UDW) detectors separated by proper distance  $L$ , both of which are at rest relative to this coordinate system. When the detectors are in motion, we have to replace  $t - T$  with  $t(\tau) - t(T)$ , where  $t$  is Minkowski coordinate time and  $\tau$  and  $T$  are the proper times of each detector, which take the roles of  $t$  and  $T$  in the previous sections. Similarly, we have to replace  $\mathbf{x} - \mathbf{X}$  with  $\mathbf{x}(\tau) - \mathbf{x}(T)$ , where now  $\mathbf{x}$  takes the role of Minkowski spatial coordinates. In this case, in general it is not true that  $\tau$  is a simple function of  $T$ , so it may not always be possible to find an upper limit version similar to equation (12). The static detectors (at rest relative to Minkowski coordinates) correspond to the special case where  $t(\tau) - t(T) = \tau - T$ , which can then be rewritten as  $t - T$ . This complexity occurs more generally when the spacetime is not flat, such as when we consider two-dimensional truncation of Schwarzschild spacetime [5].

For this reason, it is useful to try to integrate equation (10) using the Heaviside step function directly. Since the built-in function `HeavisideTheta` is not defined for complex numbers, we approximate it with a smooth function. For example, the Heaviside step function is a limit of a logistic function:

$$\theta(z) = \lim_{k \rightarrow \infty} \frac{1}{2} \times (1 + \tanh(kz)), \tag{14}$$

provided that we define  $\theta(0) = \frac{1}{2}$ . We can thus define an approximate Heaviside step function for any fixed  $k > 0$  by

$$\theta_k(z) := \frac{1}{2} \times (1 + \tanh(kz)). \tag{15}$$

Any other function used in computing a cumulative probability distribution would work.

Our integral  $K_2$  now becomes

$$K_3(\Omega, \sigma, k) := - \int_{\mathbb{R}} \int_{\gamma(\epsilon)} e^{\frac{-t^2}{2\sigma^2}} e^{\frac{-T^2}{2\sigma^2}} e^{i\Omega(t+T)} (\theta_k(t-T) W(t, \mathbf{x}; T, \mathbf{X}; \epsilon) + \theta_k(T-t) W(T, \mathbf{X}; t, \mathbf{x}; \epsilon)) dt dT, \tag{16}$$

again with the rectangular contour of Figure 2. The task now is to manually find  $k$  and  $\epsilon$  that yield good results by trial and error. Again we set  $\mathbf{X} - \mathbf{x} = (X - x, 0, 0) = (L, 0, 0)$  so that  $|\mathbf{X} - \mathbf{x}| = L$ .

We define  $\theta$  and  $K3$ .

$$\theta[\mathbf{k}_-, \mathbf{z}_-] := \frac{1}{2} \times (1 + \text{Tanh}[\mathbf{k} \mathbf{z}])$$

$$\begin{aligned} K3[\Omega_-, \sigma_-, \epsilon_-, \mathbf{k}_-, \mathbf{x}_-, \mathbf{X}_-, \text{tmax}_-, \text{minrecursion}_-] := \\ -\text{Quiet}[\text{NIntegrate}[\text{Exp}\left[-\frac{\mathbf{t}^2}{2\sigma^2}\right] \times \text{Exp}\left[-\frac{\mathbf{T}^2}{2\sigma^2}\right] \times \text{Exp}[\mathbf{I} \Omega (\mathbf{t} + \mathbf{T})] \times \\ \theta[\mathbf{k}, \mathbf{t} - \mathbf{T}] \left(-\frac{1}{4\pi^2} \frac{1}{(\mathbf{t} - \mathbf{T})^2 - (\mathbf{x} - \mathbf{X})^2}\right), \{\mathbf{t}, -\text{tmax}, \text{tmax}\}, \\ \{\mathbf{T}, -\text{tmax}, -\text{tmax} + \epsilon \mathbf{I}, \text{tmax} + \epsilon \mathbf{I}, \text{tmax}\}, \\ \text{MinRecursion} \rightarrow \text{minrecursion}]] - \\ \text{Quiet}[\text{NIntegrate}[\text{Exp}\left[-\frac{\mathbf{t}^2}{2\sigma^2}\right] \times \text{Exp}\left[-\frac{\mathbf{T}^2}{2\sigma^2}\right] \times \text{Exp}[\mathbf{I} \Omega (\mathbf{t} + \mathbf{T})] \times \\ \theta[\mathbf{k}, \mathbf{T} - \mathbf{t}] \left(-\frac{1}{4\pi^2} \frac{1}{(\mathbf{T} - \mathbf{t})^2 - (\mathbf{x} - \mathbf{X})^2}\right), \{\mathbf{T}, -\text{tmax}, \text{tmax}\}, \\ \{\mathbf{t}, -\text{tmax}, -\text{tmax} + \epsilon \mathbf{I}, \text{tmax} + \epsilon \mathbf{I}, \text{tmax}\}, \\ \text{MinRecursion} \rightarrow \text{minrecursion}]]] \end{aligned}$$

The results are promising, since there is a range of  $\epsilon$  that gives a result very close to the benchmark  $K_1 = -0.0248507 + 0.0404108 i$ , but we can do better.

```
Module[{ $\Omega = 1$ ,  $\sigma = 1$ ,  $L = 1$ ,  $\text{tmax} = 5$ ,  $\text{minrecursion} = 3$ },
  TableForm[
    ParallelTable[
      { $10 \cdot 0^{-\epsilon}$ ,  $K3[\Omega, \sigma, 10^{-\epsilon}, 5, 0, L, \text{tmax}, \text{minrecursion}]$ },
      { $\epsilon$ , -1, 5}]]]
]
```

```
10.      1.5052  $\times 10^7$  - 1.77699  $\times 10^6$  i
1.        -0.0248507 + 0.00731934 i
0.1      -0.0248507 + 0.0404071 i
0.01     -0.0248507 + 0.0404071 i
0.001    -0.0248416 + 0.0403968 i
0.0001   -0.0244954 + 0.0300156 i
0.00001  -0.0343124 + 0.0127622 i
```

The key is to set  $k \approx \epsilon^{-1}$  so that  $k \epsilon = O(1)$ . For example, for  $k = 5$ , a good choice of  $\epsilon$  is  $1/5$  but not  $1$  or  $1/12$ . The only difference in the next command from the last is in the fourth argument of `K3`, where `k` is  $10^\epsilon$  instead of  $5$ . The results are slightly better.

```
Module[{Ω = 1, σ = 1, L = 1, tmax = 5, minrecursion = 3},
  TableForm[
    ParallelTable[
      {10.0-ε, K3[Ω, σ, 10-ε, 10ε, 0, L, tmax, minrecursion]}},
      {ε, -1, 5}]]
]
```

10.	$9.43585 \times 10^6 + 6.25012 \times 10^7 i$
1.	$-0.0248507 + 0.0307766 i$
0.1	$-0.0248507 + 0.0404108 i$
0.01	$-0.0248507 + 0.0404108 i$
0.001	$-0.0248462 + 0.0404048 i$
0.0001	$-0.0248537 + 0.029866 i$
0.00001	$-0.0367152 + 0.0189248 i$

We can study this relationship better for a smaller set of values of  $k$  and judge the quality.

```
Module[{Ω = 1, σ = 1, L = 1, tmax = 5, minrecursion = 3},
  TableForm[
    ParallelTable[
      {k, K3[Ω, σ, k-1, k, 0, L, tmax, minrecursion]}},
      {k, 5, 21, 2}]]
]
```

5	$-0.0248507 + 0.0404071 i$
7	$-0.0248507 + 0.0404107 i$
9	$-0.0248507 + 0.0404108 i$
11	$-0.0248507 + 0.0404108 i$
13	$-0.0248507 + 0.0404108 i$
15	$-0.0248507 + 0.0404108 i$
17	$-0.0248507 + 0.0404108 i$
19	$-0.0248507 + 0.0404108 i$
21	$-0.0248507 + 0.0404108 i$

Thus we have obtained a good range of  $\epsilon$  where the deformation theorem clearly works. We need to work with  $k \epsilon = 1$  because the smooth approximation of the Heaviside step functions introduces infinitely many new poles in the complex plane, since  $\theta_k(z)$  has (countably) infinitely many poles along the imaginary axis. This can be easily seen by rewriting  $\theta_k$  as:

$$\theta_k(z) = \frac{1}{2} + \frac{1}{2} \tanh(kz) = \frac{1}{1 + e^{-2kz}}. \tag{17}$$

Therefore the choice of  $k$  must depend on  $\epsilon$  so that the contour does not cross extra poles, making the integral gain spurious contributions. Nonetheless, finding the right  $k$  that works is an empirical process, especially when the Wightman function is very complicated. Our examples thus far are simple enough that they can be easily cross-validated and evaluated without much effort.

## ■ Further Applications

The above examples have been used by the author to calculate the density matrix elements of Unruh–DeWitt detectors interacting with massless scalar fields for various background spacetimes, where the calculations require time-dependent perturbation theory involving Dyson series expansions. Starting from the calculations in [4] where a primitive version was first used to study entanglement harvesting in two-dimensional spacetimes with accelerating mirror (see the Appendix in [4]), the technique was subsequently improved (using a full rectangular contour that is not close to the poles) for the entanglement harvesting problem in Schwarzschild and collapsing shell spacetimes [5]. In [5], the derivative-coupling functions (strictly speaking they are not Wightman functions but the proper time derivatives of the Wightman functions) are so complicated that it is somewhat surprising the upper-limit technique outlined above worked very well across large parameter spaces. The author has verified that the approximation of the Heaviside step function works well for calculations in [5], although with the caveat: when the derivative-coupling Wightman functions can be expressed as a sum of distinct terms, the part of the derivative Wightman function that contains  $u - u'$  and  $U - U'$  requires relatively small  $k$  but the part that contains  $v - v'$  or  $V - V'$  may need very large  $k \approx 10^{-3}$ . This highlights the need for empirically testing the choice of  $k$  before proceeding with physical calculations.

We extend the results of [5] for harvesting of correlations by two quantum-mechanical detectors interacting with a quantum field in [9], where we consider more complicated detector trajectories. In this context, the authors verified that the upper-limit form analogous to equation (12) proved numerically unstable in physically relevant regimes, thus making indispensable the approximate Heaviside step functions as a way to calculate the nonlocal correlation term in entanglement harvesting contexts. In that setting, the calculation of the nonlocal term involves an integral similar to  $K_1$  in equation (10), but with much more complicated numerical calculations than in [5]; in particular the upper limits are not simply affine functions of  $\tau$  or  $\tau'$ . We will explain the technicalities specific to the problem at hand when invoking the techniques outlined in [9].

In quantum field theory in generic spacetimes (flat or curved), there are many situations of computing integrals over distributions. Our techniques should be general enough to use in such contexts, where correlation functions take a form similar to Wightman functions. For example, we have verified that our techniques work for computing field commutators [5] and in  $(2 + 1)$ -dimensional rotating BTZ black holes [10, 11]. We are also able to reproduce the computations of transition rates as done for example in [12] (also see Appendix of [4]), which are fundamentally simpler because they are one-dimensional integrals over the Wightman functions.

## ■ Conclusion

In this article we have presented a straightforward implementation of contour integration using the `Integrate` and `NIntegrate` commands, taking advantage of powerful results in complex analysis. The goal is to provide a systematic example-based implementation of contour integration that does not require any user-defined functions and only requires setting options. We provide explicit examples of how this can be used when integrating analytically and numerically some commonly used distributions, such as Wightman functions in quantum field theory. We also provide efficient approximation schemes to compute quantities involving time-ordering operations where Heaviside step functions are required. Our examples are geared toward research in relativistic quantum information, where finite-dimensional quantum systems interact with a quantum field on a possible curved background spacetime. We believe our implementation will be useful for calculations involving various Green's functions, propagators, integral kernels and so on for both students and researchers in fields requiring numerical evaluations of integrals over distributions without resorting to unnecessary or excessive approximation schemes.

## ■ Acknowledgments

The author thanks Kensuke Gallock-Yoshimura for help in testing the numerical schemes proposed here in [9], and Nicholas Funai for reading through the draft and providing useful feedback. The author acknowledges support from the Mike-Ophelia Lazaridis Fellowship from the Institute for Quantum Computing (IQC), University of Waterloo, Canada. The author also would like to thank both the reviewer and editor for very useful and clarifying comments and recommendations.

## ■ References

- [1] J. M. Howie, *Complex Analysis*, London: Springer Science & Business Media, 2003.
- [2] J. W. Brown and R. V. Churchill, *Complex Variables and Applications*, 8th ed., New York: McGraw-Hill Education, 2009.
- [3] V. Mukhanov and S. Winitzki, *Introduction to Quantum Effects in Gravity*, New York: Cambridge University Press, 2007.
- [4] E. Tjoa. "Aspects of Quantum Field Theory with Boundary Conditions," *UWSpace*, 2019. [hdl.handle.net/10012/14843](https://hdl.handle.net/10012/14843).
- [5] E. Tjoa and R. B. Mann, "Harvesting Correlations in Schwarzschild and Collapsing Shell Spacetimes," *Journal of High Energy Physics*, **08**(155), 2020. [doi.org:10.1007/JHEP08\(2020\)155](https://doi.org/10.1007/JHEP08(2020)155).
- [6] N. D. Birrell and P. C. W. Davies, *Quantum Fields in Curved Space*, Cambridge: Cambridge University Press, 1982.
- [7] A. R. H. Smith. "Detectors, Reference Frames, and Time," *UWSpace*, 2017. [hdl.handle.net/10012/12618](https://hdl.handle.net/10012/12618).

- [8] G. Ver Steeg and N. C. Menicucci, “Entangling Power of an Expanding Universe,” *Physical Review D*, **79**(4), 2009 044027. doi.org:10.1103/PhysRevD.79.044027.
- [9] K. Gallock-Yoshimura, E. Tjoa and R. B. Mann, “Harvesting Entanglement with Detectors Freely Falling into a Black Hole.” arxiv.org/abs/2102.09573.
- [10] M. P. G. Robbins, L. J. Henderson and R. B. Mann, “Entanglement Amplification from Rotating Black Holes.” arxiv.org/abs/2010.14517.
- [11] L. J. Henderson, R. A. Hennigar, R. B. Mann, A. R. H. Smith and J. Zhang, “Harvesting Entanglement from the Black Hole Vacuum,” *Classical and Quantum Gravity*, **35**(21) 21LT02, 2018. doi.org/10.1088/1361-6382/aae27e.
- [12] J. Louko and A. Satz, “Transition Rate of the Unruh–DeWitt Detector in Curved Spacetime,” *Classical and Quantum Gravity*, **25**(5) 055012, 2008. doi.org/10.1088/0264-9381/25/5/055012.

Erickson Tjoa, “Numerical Contour Integration,” *The Mathematica Journal*, 2021. doi.org/10.3888/tmj.23–3.

## About the Author

Erickson Tjoa was born in Jakarta, Indonesia. He moved to Singapore at the age of 15 and studied physics and mathematics as a double major at Nanyang Technological University from 2013-2017. He then completed an MSc at the University of Waterloo, where he is currently a second-year PhD student in relativistic quantum information.

### **Erickson Tjoa**

*Department of Physics and Astronomy  
Institute for Quantum Computing (IQC)  
University of Waterloo, ON, Canada  
200 University Avenue West  
Waterloo, ON, Canada N2L 3G1  
e2tjoa@uwaterloo.ca*